# $Id: PKGBUILD 88530 2013-04-19 10:33:08Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Benjamin Andresen <benny@klapmuetz.org>
# Contributor: Mikko Seppдlд <t-r-a-y@mbnet.fi>

pkgname=xvkbd
pkgver=3.4a
pkgrel=1
pkgdesc="virtual (graphical) keyboard program for X Window System"
arch=('i686' 'x86_64')
url="http://homepage3.nifty.com/tsato/xvkbd/"
license=('GPL')
depends=('libxmu' 'xaw3d' 'glibc' 'libxt' 'libxtst' 'libxp' 'libxpm')
makedepends=('imake')
source=(http://homepage3.nifty.com/tsato/xvkbd/xvkbd-$pkgver.tar.gz)
md5sums=('3d996ba9f84b5ed2392f9809ada22711')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  xmkmf
  sed -i 's|#include <X11/Xaw/|#include <X11/Xaw3d/|' xvkbd.c
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR=$pkgdir install
  # fix
  rm -f $pkgdir/usr/lib/X11/app-defaults
  mkdir -p $pkgdir/usr/lib/X11
  mv $pkgdir/etc/X11/app-defaults $pkgdir/usr/lib/X11/
  rm -rf $pkgdir/etc/
}
